

Computing the molecular expansion of species with the Maple package *Devmol*

Pierre Auger, Gilbert Labelle and Pierre Leroux
LaCIM, Université du Québec à Montréal

Introduction



Intuitively, the molecular expansion of a species F is a classification of its structures according to their size and symmetries. It can be seen as a refinement of the *tilde* generating function

$$\tilde{F}(x) = \sum_{n \geq 0} \tilde{f}_n x^n \quad (1)$$

of the number \tilde{f}_n of unlabelled F -structures of size n .

For example, the tilde generating function $\tilde{A}(x)$ of the species A of (free) rooted trees, up to size 6, is

$$\tilde{A}(x) = x + x^2 + 2x^3 + 4x^4 + 9x^5 + 20x^6 + \dots \quad (2)$$

It is refined by the molecular expansion of A given in Section 1.1 (see (6)) where each unlabelled rooted tree is represented by a term (a molecular species) describing its symmetries. For example, for degree 3, the term X^3 represents the rooted chain  while $X E_2(X)$ represents the rooted star .

The goal of the Maple package *Devmol* is to compute such molecular expansions, up to a specified degree, for species of the so-called *cyclo-set type*

(in French: espèces cyclo-ensemblistes). These are species which have either an explicit or a recursive definition in terms of the species L , of lists, E , of sets, and C , of oriented cycles. More precisely, since we are dealing with species of bounded degrees, the specifications will be in terms of the species L_n , E_n , and C_n , for $n \geq 0$, where the index n denotes the restriction to degree n . Note that $L_0 = E_0 = 1$, $C_0 = 0$, $L_1(X) = E_1(X) = C_1(X) = X$, $C_2 = E_2$ and $L_n(X) = X^n$.

Moreover, *Devmol* has the capacity to handle weighted multisort species with scalar coefficients in the complex field \mathbb{C} . The sort variables have to be declared, except X, Y, Z which are sort variables by default, and the weights have to be monomials (commutative words) in some indeterminates which must be declared as weight variables, except t which is a weight variable, by default. All other unspecified variables or “names” are considered as complex scalars.

The reader is referred to [9] and [2] for a basic presentation of species theory and also to [4] where some molecular expansions are computed using Maple. We review here some of the notions related to molecular species. Two structures of a species F are *isomorphic* if one can be obtained from the other by a relabelling along a bijection. *Unlabelled* F -structures can be seen as isomorphism classes of F -structures. A species M is called *molecular* if it has only one isomorphism class. It is completely determined by the stabilizer $H = \text{Stab}(s)$ of one of its structures, say $s \in M[n]$, where n is the degree of M and $[n] = \{1, 2, \dots, n\}$. We write $M(X) = X^n/H$. In particular, we have

$$L_n = X^n/\{1\}, \quad E_n = X^n/S_n, \quad C_n = X^n/\langle \rho \rangle, \quad (3)$$

where S_n denotes the symmetric group on $[n]$ and ρ generates a cyclic subgroup of order n of S_n .

For a general molecular species

$$M(X) = X^n/H,$$

the cycle index series $Z_M(x_1, x_2, \dots)$ coincides with the cycle index polynomial $Z_H(x_1, x_2, \dots)$ of $H \leq S_n$. Molecular species are closed under the basic operations of product and (partitional) composition of species, corresponding to the operations of product and of wreath product of their stabilizers, respectively. See [2], Section 4.6, for more details.

Two F -structures are called *similar* if they have conjugate stabilizers. It is equivalent to say that one structure can be relabelled in such a way that they have exactly the same stabilizer. This corresponds to isomorphic molecular species. Hence similar (but non-isomorphic) F -structures give rise to multiplicities in the molecular expansion of F . For example, the three rooted trees shown in the first row of Figure 1 are not isomorphic but they are similar since they are of the type $X^3E_2(X)$, having only a two-element group generated by a transposition as stabilizers. This explains the multiplicity 3 in the term $3X^3E_2(X)$. Similarly, the three rooted trees in the second row are similar since they are asymmetric, giving rise to the term $3X^5$ in the molecular expansion of the species A of rooted trees.

The *molecular expansion* of a species F is defined as the equality (natural isomorphism)

$$F = \sum_{M \in \mathcal{M}} f_M M, \quad (4)$$

where f_M is the multiplicity of the molecular species M in F and \mathcal{M} is an appropriate set of molecular species.

In the following section, we present two examples of molecular expansions: for rooted trees, and for Husimi graphs weighted by monomials describing their block-size distributions. In Section 2, we describe briefly the addition formulas for the basic molecular species X^n , E_n , and C_n , which are essential for the computation of molecular expansions. In Section 3, we show how these formulas can be used to compute recursively the molecular expansion of the species of R -enriched rooted trees, when R itself is of cyclo-set type, and also of R -enriched trees, using the Dissymmetry Theorem for trees.

The Maple package Devmol was developed by Pierre Auger at LaCIM for the purpose of computing molecular expansions. We give a detailed description of the package in Section 4, and conclude with two examples of applications, in Section 5, to rooted trees weighted by a leaf counter and to connected graphs, all of whose blocks are in a given family \mathcal{B} of species of cyclo-set type.

1 Some examples of molecular expansions

The *Devmol* package is well adapted to compute the molecular expansion of tree-like species. Two examples are given in this section: rooted trees and Husimi graphs.

1.1 Rooted trees

First, consider the well-known species A of rooted trees, defined by the functional equation

$$A = XE(A), \tag{5}$$

where E denotes the species of “sets” (E for French “ensembles”). Its molecular expansion can be seen as an explicit description of the species. Up to degree (size) 6, we have

$$\begin{aligned} A = & X + X^2 + XE_2(X) + X^3 + XE_3(X) + 2X^4 + X^2E_2(X) \\ & + XE_4(X) + 3X^3E_2(X) + XE_2(X^2) + 3X^5 + X^2E_3(X) \\ & + X^2E_4(X) + 6X^4E_2(X) + 2X^2E_2(X^2) + 3X^3E_3(X) \\ & + X^2E_2(X)^2 + XE_5(X) + 6X^6 + \dots \end{aligned} \tag{6}$$

The homogeneous component of degree 5 of this molecular expansion is

$$XE_4(X) + 3X^3E_2(X) + XE_2(X^2) + 3X^5 + X^2E_3(X)$$

These terms are illustrated in Figure 1.

1.2 Weighted Husimi graphs

We illustrate now the more general context, in which weighted species occur. The weight of a structure is assumed to be a unitary monomial (a commutative word) in some given indeterminates, called *weight variables*. For example, let the weight variables be y_2, y_3, \dots . Define a *Husimi graph* as a connected graph all of whose blocks (2-connected components) are complete graphs (see [14]). The weight $w(g)$ of a Husimi graph g is defined by $w(g) = y_2^{n_2} y_3^{n_3} \dots$, where n_i is the number of blocks of g of size i for




Figure 1: Rooted trees of size 5

$i = 2, 3, \dots$. An example is given in Figure 2. The weighted species of Husimi graphs is denoted by Hu_w . Its molecular expansion up to degree 6 is computed by *Devmol* as:

Degree 1:

$$X$$

Degree 2:

$$y_2 E_2(X)$$

Degree 3:

$$y_2^2 X E_2(X) + y_3 E_3(X)$$

Degree 4:

$$y_2 y_3 X^2 E_2(X) + y_2^3 X E_3(X) + y_2^3 E_2(X^2) + y_4 E_4(X)$$

Degree 5:

$$(y_3 y_2^2 + y_2^4) X^3 E_2(X) + y_2^2 y_3 X E_2(X)^2 + (y_3 y_2^2 + y_2^4) X E_2(X^2) + y_2^4 X E_4(X) + y_3^2 X E_2(E_2(X)) + y_5 E_5(X) + y_2 y_4 X^2 E_3(X)$$




Figure 2: A Husimi graph of weight $y_2^4 y_3 y_4^2 y_5$

Degree 6:

$$\begin{aligned}
& y_2^5 X^2 E_2(X^2) + (y_2^2 y_4 + y_3 y_4 + y_3 y_2^3) X E_2(X) E_3(X) + y_2^2 y_4 E_2(X) E_2(X^2) \\
& + (y_3^2 y_2 + y_2^5 + 3 y_3 y_2^3) X^4 E_2(X) + y_2^5 E_2(X^3) + (y_3^2 y_2 + y_2^5) E_2(X E_2(X)) \\
& + y_2 y_3^2 X^2 E_2(E_2(X)) + y_3 y_2^3 E_3(X^2) + y_2 y_5 X^2 E_4(X) + y_6 E_6(X) \\
& + y_3 y_2^3 X^2 E_2(X)^2 + y_3 y_2^3 X^6 + (y_2^5 + y_2^2 y_4) X^3 E_3(X) + y_2^5 X E_5(X).
\end{aligned}$$

As an illustration, the nine terms of size 5 are shown in Figure 3. Note that in terms of simple graphs, the species E_n can be interpreted both as that of discrete graphs or of complete graphs, on n vertices.

2 Addition formulas

The main feature of *Devmol* is its capacity to implement the addition formulas relative to the species of sets and of cycles. Molecular expansions are then computed by applying recursively these formulas. In what follows, the subscript n in E_n and C_n denotes the restriction to degree (cardinality)










 $y_2^4 \cdot X^3 E_2(X)$	 $y_2^2 y_3 \cdot X^3 E_2(X)$	 $y_2^2 y_3 \cdot X E_2(X)^2$
 $y_2^4 \cdot X E_2(X^2)$	 $y_2^2 y_3 \cdot X E_2(X^2)$	 $y_2^4 \cdot X E_4(X)$
 $y_3^2 \cdot X E_2(E_2(X))$	 $y_2 y_4 \cdot X^2 E_3(X)$	 $y_5 \cdot E_5(X)$

Figure 3: Husimi graphs of size 5

n . Observe that Maple already knows the addition formulas for the species $L_n = X^n$, using the “expand” function: for a given $n \geq 0$, we have, for example,

$$(X + Y)^n = \sum_{i=0}^n \binom{n}{i} X^i Y^{n-i}. \quad (7)$$

Note that for any molecular species M of degree n , we have $M(1) = 1$ and, more generally, for any scalar k ,

$$M(k) = Z_M(k, k, \dots). \quad (8)$$

Formula (8) holds for any non-negative integer k in virtue of Pólya theory and the extension to a general scalar is valid by polynomiality. Also, for any

weight variable t , we have

$$M(tX) = t^n M(X). \quad (9)$$

This applies, in particular, to the molecular species $M = L_n, E_n$ and C_n .

The additions formulas given in the following subsections are taken from [1]. They are implemented in Devmol.

2.1 Addition formulas related to the species of sets

Recall that $E_0(X) = 1$, the “empty set” species, and that $E_1(X) = X$. For $n, k \geq 0$, we have

$$E_n(X_1 + \cdots + X_k) = \sum_{\substack{n_1 + \cdots + n_k = n \\ n_i \geq 0}} E_{n_1}(X_1) \cdots E_{n_k}(X_k). \quad (10)$$

It follows that

$$E_n(kX) = \sum_{i=1}^n \binom{k}{i} \sum_{\lambda \vdash_i n} \binom{n}{\lambda_1, \dots, \lambda_i} E_{\lambda_1}(X) \cdots E_{\lambda_i}(X), \quad (11)$$

the expression $\lambda \vdash_i n$ meaning that the summation index λ is a partition of n into i parts, and that

$$E_n(k + X) = \frac{k^{\langle n \rangle}}{n!} + \frac{k^{\langle n-1 \rangle}}{(n-1)!} X + \cdots + E_n(X), \quad (12)$$

where $k^{\langle n \rangle} = k(k+1) \cdots (k+n-1)$.

2.2 Addition formulas related to the species of cycles

For the species C of oriented cycles, we have

$$C_n(X_1 + \cdots + X_k) = \sum_{d|n} \sum_{\substack{\psi_1 + \cdots + \psi_k = d \\ \psi_i \geq 0}} \lambda_\psi C_{n/d} \left(X_1^{\psi_1} \cdots X_k^{\psi_k} \right), \quad (13)$$

where, for $\psi = (\psi_1, \dots, \psi_k)$, λ_ψ denotes the number of Lyndon words on $\mathfrak{X} = \{x_1 < \dots < x_k\}$ having distribution ψ , i. e. ψ_ν occurrences of the letter x_ν , $\nu = 1, \dots, k$. Recall that a *Lyndon word* is a primitive (not a power of another word) lexicographically minimal word in its cyclic conjugacy class. Using Möbius inversion, it can be shown that

$$\lambda_\psi = \frac{1}{\psi_1 + \dots + \psi_k} \sum_{d|\gcd(\psi_1, \dots, \psi_k)} \mu(d) \left(\frac{\psi_1 + \dots + \psi_k}{d}, \frac{\psi_1}{d}, \dots, \frac{\psi_k}{d} \right). \quad (14)$$

Moreover, one has

$$C_n(kX) = \sum_{d|n} \lambda_d(k) C_{n/d}(X^d), \quad (15)$$

where $\lambda_d(k) = \frac{1}{d} \sum_{\delta|d} \mu(\delta) k^{d/\delta}$ is the number of Lyndon words of length d in a k -element alphabet, and

$$C_n(k + X) = \frac{1}{n} \sum_{d|n} \phi(d) k^{n/d} + \sum_{d|n} \sum_{\substack{i+j=d \\ i>0}} \lambda_{i,j}(k) C_{n/d}(X^i), \quad (16)$$

the coefficients $\lambda_{i,j}(k)$ being given by the formula

$$\lambda_{i,j}(k) = \frac{1}{i+j} \sum_{\delta|(i,j)} \mu(\delta) \binom{(i+j)/\delta}{i/\delta} k^{j/\delta}. \quad (17)$$

Note that by polynomiality, equations (11), (12), (15) and (16) are also valid for any scalar $k = \xi \in \mathbb{C}$.

3 Species implicitly defined by functional equations

3.1 R -enriched rooted trees

Let R be a species such that $R(0) \neq 0$. The species A_R of *R -enriched rooted trees* is the unique species satisfying the combinatorial equation

$$A_R = X R(A_R). \quad (18)$$




Figure 4: Typical R -enriched rooted tree

Figure 4 shows a typical R -enriched rooted tree on the set $U = \{a, b, \dots, k\}$, where all edges have been oriented towards the root, the vertex e . Hence, an A_R -structure is a rooted tree where the *fiber* of each vertex (that is the set of immediate predecessors) is endowed with an R -structure. Familiar examples include

1. Ordered (or plane) rooted trees, where $R = L$,
2. Ordinary rooted trees, where $R = E$,
3. Mobiles, where $R = 1 + C$.

Let us illustrate the iterative method for computing the molecular expansion truncated to degree n , first in the case of the species A of E -enriched (i.e. ordinary) rooted trees. For any species F , we write

$$F_{\leq n} = F|_{\leq n} = F_0 + F_1 + \dots + F_n \quad (19)$$

to denote the restriction of F to sets of size at most n .

We start with

$$A_{\leq 0} = 0,$$

since there is no rooted tree on the empty set. Next, we have

$$A_{\leq 1} = (X \cdot E_0(A))|_{\leq 1} = X.$$

The general iteration formula is given by

$$\begin{aligned} A_{\leq n} &= (X \cdot E_{\leq n-1}(A_{\leq n-1}))|_{\leq n} \\ &= X \cdot E_{\leq n-1}(A_{\leq n-1})|_{\leq n-1}. \end{aligned} \tag{20}$$

Hence, one has successively

$$\begin{aligned} A_{\leq 2} &= X \cdot E_{\leq 1}(A_1)|_{\leq 2} \\ &= X \cdot ((1 + X) \circ X)|_{\leq 2} \\ &= X \cdot (1 + X) \\ &= X + X^2, \end{aligned}$$

$$\begin{aligned} A_{\leq 3} &= X \cdot E_{\leq 2}(A_{\leq 2})|_{\leq 3} \\ &= X \cdot ((1 + X + E_2(X)) \circ (X + X^2))|_{\leq 3} \\ &= X \cdot (1 + X + X^2 + E_2(X + X^2))|_{\leq 3} \\ &= X \cdot (1 + X + X^2 + E_2(X) + X^3 + E_2(X^2))|_{\leq 3} \\ &= X + X^2 + X^3 + X E_2(X), \end{aligned}$$

and so on.

Consider now the general case of the species $A = A_R$ of R -enriched rooted trees, solution of the combinatorial equation $A = XR(A)$. We start with $A_0 = 0$ or $A_1 = XR_0$. Furthermore, the general recurrence relation can be written as

$$A_{\leq n} = X \cdot R_{\leq n-1}(A_{\leq n-1})|_{\leq n-1}. \tag{21}$$

The molecular expansion of A can then be computed recursively, using (21). Hence, the Maple package *Devmol* can compute the molecular expansion of the species A_R for any R of cyclo-set type. For example, the molecular

expansion of the species A_{1+C} of “mobiles” is, up to degree 6,

$$\begin{aligned}
 A_{1+C}|_{\leq 6} = & X + X^2 + XE_2(X) + X^3 + XC_3(X) + 2X^4 + X^2E_2(X) \\
 & + XC_4(X) + 4X^5 + XE_2(X^2) + 2X^3E_2(X) + X^2C_3(X) \\
 & + XC_5(X) + 5X^4E_2(X) + 2X^3C_3(X) + X^2C_4(X) \quad (22) \\
 & + X^2E_2(X^2) + 10X^6.
 \end{aligned}$$

3.2 R -enriched trees.

R -enriched trees are defined as (unrooted) trees where the set of neighbors of any vertex is endowed with an R -structure. See Figure 5 for an illustration. For example, plane trees can be seen as $1 + C$ -enriched trees.

The species of R -enriched trees is denoted by a_R . It can be expressed in terms of the species $A_{R'}$ of R' -enriched rooted trees, where R' denotes the derivative of the species R , using the Dissymmetry Theorem for R -enriched trees (see [2], Theorem 4.1.2). Indeed, we have

$$a_R = XR(A_{R'}) + E_2(A_{R'}) - A_{R'}^2. \quad (23)$$




Figure 5: Typical R -enriched tree




Figure 6: Typical G -additive rooted tree

3.3 G -additive rooted trees.

Another important family of species defined by functional equations is that of G -additive rooted trees, where G is a given species such that $\mathcal{O}(G) \geq 2$, that is $G(0) = G'(0) = 0$. It is defined as the species $B = B(X)$ which satisfies the additive functional equation

$$B = X + G(B). \quad (24)$$

This species is often denoted by $B = {}_G A$. See Figure 6 for an illustration of a typical ${}_G A$ -structure.

An example is also given by the species $P = P(X)$ of parenthesizations, which can be defined by the functional equation

$$P = X + P^2. \quad (25)$$

It is clear that the molecular expansion of the species ${}_G A$ of G -additive rooted trees can also be computed recursively, whenever effective addition formulas are known for G .

4 Presentation of the Maple package *Devmol*

4.1 Basic purpose

The basic purpose of the Maple package *Devmol* is to compute the molecular expansion of any given recognized species expression, extending the expansion of classical algebraic expressions. For instance, the molecular expansion of the expression $(X + Y)^2$ is simply $X^2 + 2XY + Y^2$.

Let \mathcal{A} denote the alphabet of sort variables and \mathcal{V} the alphabet of weight variables. The language $\mathcal{L}_{\mathcal{A}}$ of recognized expressions is then defined by the following grammar :

1. $1 \in \mathcal{L}_{\mathcal{A}}$,
2. $X \in \mathcal{A} \Rightarrow X \in \mathcal{L}_{\mathcal{A}}$,
3. $p \in \mathbb{C}[\mathcal{V}], \Phi \in \mathcal{L}_{\mathcal{A}} \Rightarrow p * \Phi \in \mathcal{L}_{\mathcal{A}}$,
4. $\Phi, \Psi \in \mathcal{L}_{\mathcal{A}} \Rightarrow \Phi + \Psi \in \mathcal{L}_{\mathcal{A}}$ and $\Phi * \Psi \in \mathcal{L}_{\mathcal{A}}$,
5. $n \in \mathbb{N}, \Phi \in \mathcal{L}_{\mathcal{A}} \Rightarrow E_n(\Phi) \in \mathcal{L}_{\mathcal{A}}$ and $C_n(\Phi) \in \mathcal{L}_{\mathcal{A}}$.

The package *Devmol*, or , more precisely, its main procedure `dev`, can then be seen as an operator of the form

$$\mathcal{L}_{\mathcal{A}} \longrightarrow \mathbb{C}[\mathcal{V}][[\mathcal{M}_{\mathcal{A}}]]$$

which computes the molecular expansion of a recognized expression. The result is a formal power series with coefficients in the ring $\mathbb{C}[\mathcal{V}]$ over the set $\mathcal{M}_{\mathcal{A}}$, of (unweighted) molecular species of cyclo-set type in the sorts \mathcal{A} . This set is defined recursively as follows:

1. $1 \in \mathcal{M}_{\mathcal{A}}$
2. $X \in \mathcal{A} \Rightarrow X \in \mathcal{M}_{\mathcal{A}}$,
3. $M, N \in \mathcal{M}_{\mathcal{A}} \Rightarrow M \cdot N \in \mathcal{M}_{\mathcal{A}}$,

4. $n \in \mathbb{N}, n \geq 2, M \in \mathcal{M}_{\mathcal{A}} \Rightarrow E_n(M) \in \mathcal{M}_{\mathcal{A}},$
5. $n \in \mathbb{N}, n \geq 3, M \in \mathcal{M}_{\mathcal{A}} \Rightarrow C_n(M) \in \mathcal{M}_{\mathcal{A}}.$

The Maple package *Devmol* comprises basically two main (internal) procedures, `expand/E_` and `expand/C_` which extend the Maple procedure `expand`. When the `expand` procedure is applied to a recognized expression, it recursively calls the relevant procedure each time a sub-expression of the form $E_n(\cdot)$ or $C_n(\cdot)$ is met. Each of these procedures “executes” the corresponding addition formula.

As mentioned in the introduction, the default set of sort variables is $\mathcal{A} = \{X, Y, Z\}$. To include new sort variables, one uses the procedure `ajoutvs`, which is short for French “ajouter variables de sortes” (English alias: `addsv`). More precisely, the command

```
> ajoutvs(YY);
```

adds the name *YY* to the set \mathcal{A} of sort variables. Maple also answers by printing the current set of sort variables. Note that the sort variables are global. In the same manner, one uses the procedure `ajoutvv` (English alias: `addvv`) to add new elements to the set \mathcal{V} of weight variables. Once added to this set, a weight variable becomes global too. The initial set of weight variables is $\mathcal{V} = \{t\}$.

4.2 The procedure `dev`

The most fundamental procedure of *Devmol* is `dev`. The purpose of `dev(Phi,d)` is to compute the molecular expansion of the recognized expression *Phi*, up to the given truncation degree *d*.

Examples.

```
> ajoutvv(u);
                                {t, u}
> Phi := E[5](E[2](X) + u*X);
                                Φ := E5(E2(X) + u X)
> dev(Phi,10);
```

```

u^5 E_5(X) + E_5(E_2(X)) + u^4 E_4(X) E_2(X) + u X E_4(E_2(X))
+ u^3 E_3(X) E_2(E_2(X)) + u^2 E_2(X) E_3(E_2(X))
> dev((X^2 + t*E[3](Y))^5, infinity);

X^10 + 5 t E_3(Y) X^8 + 10 t^2 E_3(Y)^2 X^6 + 10 t^3 E_3(Y)^3 X^4
+ 5 t^4 E_3(Y)^4 X^2 + t^5 E_3(Y)^5
> dev((1 - Z - Z^2)^(-1), 10);

1 + Z + 2 Z^2 + 3 Z^3 + 5 Z^4 + 8 Z^5 + 13 Z^6 + 21 Z^7 + 34 Z^8 + 55 Z^9
+ 89 Z^10
> dev(E[5](1 + t * X), 5);

1 + t X + t^2 E_2(X) + t^3 E_3(X) + t^4 E_4(X) + t^5 E_5(X)
> dev(E[5](3 + t*X), 5);

21 + 15 t X + 10 t^2 E_2(X) + 6 t^3 E_3(X) + 3 t^4 E_4(X) + t^5 E_5(X)
> dev(E[5](a + t*X), 5);

a(a+1)(a+2)(a+3)(a+4) / 120 + a(a+1)(a+2)(a+3)tX / 24
+ 1/6 a(a+1)(a+2)t^2 E_2(X) + 1/2 a(a+1)t^3 E_3(X)
+ a t^4 E_4(X) + t^5 E_5(X)
> dev(E[5](u + t*X), 5);

u^5 + u^4 t X + u^3 t^2 E_2(X) + u^2 t^3 E_3(X) + u t^4 E_4(X) + t^5 E_5(X)

```

4.3 Procedures for computing generating series

The next family of procedures is used for computing various generating series. There are three such procedures, namely `sgf(F,d)` (English alias: `egs`), which computes the exponential generating series of the species F , truncated

at degree d , `sgef` (English: `fegs`), where the counting coefficients can be directly read off, and `sgt` (English: `tgs`), which computes the (isomorphism) type generating series, i.e. the tilde generating series, for unlabelled enumeration.

Examples 1. We compute these series up to degree 10 for the species of oriented cycles.

```
> FF := sum(C[k](X), k = 1..10);
```

$$FF := C_1(X) + C_2(X) + C_3(X) + C_4(X) + C_5(X) + C_6(X) + C_7(X) \\ + C_8(X) + C_9(X) + C_{10}(X)$$

```
> GG := dev(FF, 10);
```

$$GG := X + E_2(X) + C_3(X) + C_4(X) + C_5(X) + C_6(X) + C_7(X) + C_8(X) \\ + C_9(X) + C_{10}(X)$$

```
> sgef(GG, 10);
```

$$x + \frac{1}{2}x^2 + \frac{1}{3}x^3 + \frac{1}{4}x^4 + \frac{1}{5}x^5 + \frac{1}{6}x^6 + \frac{1}{7}x^7 + \frac{1}{8}x^8 + \frac{1}{9}x^9 + \frac{1}{10}x^{10}$$

```
> sgef(GG, 10);
```

$$\frac{x}{1!} + \frac{x^2}{2!} + \frac{2x^3}{3!} + \frac{6x^4}{4!} + \frac{24x^5}{5!} + \frac{120x^6}{6!} + \frac{720x^7}{7!} + \frac{5040x^8}{8!} \\ + \frac{40320x^9}{9!} + \frac{362880x^{10}}{10!}$$

```
> sgt(GG, 10);
```

$$x + x^2 + x^3 + x^4 + x^5 + x^6 + x^7 + x^8 + x^9 + x^{10}$$

2. Here is an example involving a two-sort species:

```
> Phi := (1 + X + E[2](X) + E[3](X)) * (1 + C[3](Y));
```

$$\Phi := (1 + X + E_2(X) + E_3(X)) (1 + C_3(Y))$$

```
> dev(Phi, 9);
```

$$1 + C_3(Y) + X + X C_3(Y) + E_2(X) + E_2(X) C_3(Y) + E_3(X) \\ + E_3(X) C_3(Y)$$

```
> sge(Phi,9);
```

$$1 + \frac{1}{3}y^3 + x + \frac{1}{3}xy^3 + \frac{1}{2}x^2 + \frac{1}{6}x^2y^3 + \frac{1}{6}x^3 + \frac{1}{18}x^3y^3$$

```
> sgef(Phi,9);
```

$$\frac{1}{1!} + \frac{2y^3}{3!} + \frac{x}{1!} + \frac{2xy^3}{1!3!} + \frac{x^2}{2!} + \frac{2x^2y^3}{2!3!} + \frac{x^3}{3!} + \frac{2x^3y^3}{(3!)^2}$$

```
> sgt(Phi,9);
```

$$1 + y^3 + x + xy^3 + x^2 + x^2y^3 + x^3 + x^3y^3$$

4.4 Procedures for computing the molecular expansion of tree-like species

There are procedures in *Devmol* for computing the molecular expansion of the classes of tree-like species described in Section 3. These are `arborr` (English: `rrtrees`), for the species of R -enriched rooted trees, `arbrer` (`rtrees`), for R -enriched trees, and `arborgadd` (`gaddrtrees`), for G -additive rooted trees.

In each case, the parameters R and G are one-sort species of cyclo-set type.

4.4.1 R -enriched rooted trees

The command `arborr(R,d)` returns the molecular expansion up to degree d of the species A_R of R -enriched rooted trees.

Example. Ordinary rooted trees.

```
> R := 1 + sum(E[k](X), k = 1..5);
```

$$R := 1 + E_1(X) + E_2(X) + E_3(X) + E_4(X) + E_5(X)$$

```
> A := arborr(R,6);
```

$$\begin{aligned}
A := & X + X^2 + X E_2(X) + X^3 + X E_3(X) + 2 X^4 + X^2 E_2(X) \\
& + X E_4(X) + 3 X^3 E_2(X) + X E_2(X^2) + 3 X^5 + X^2 E_3(X) + 6 X^6 \\
& + X^2 E_2(X)^2 + X^2 E_4(X) + 6 X^4 E_2(X) + 3 X^3 E_3(X) \\
& + 2 X^2 E_2(X^2) + X E_5(X)
\end{aligned}$$

> sge(A,6);

$$x + \frac{125}{24} x^5 + \frac{3}{2} x^3 + \frac{54}{5} x^6 + \frac{8}{3} x^4 + x^2$$

> sgef(A,6);

$$\frac{x}{1!} + \frac{2x^2}{2!} + \frac{9x^3}{3!} + \frac{64x^4}{4!} + \frac{625x^5}{5!} + \frac{7776x^6}{6!}$$

> sgt(A,6);

$$x + x^2 + 2x^3 + 4x^4 + 9x^5 + 20x^6$$

4.4.2 R -enriched trees

The command `arbrer(R,d)` returns the molecular expansion up to degree d of the species a_R of R -enriched trees.

Example. The following example gives the molecular expansion (up to degree 8) of the species a_R , where $R = C$. This is the species of *plane trees*.

> R := 1 + X + sum(C[k](X), k = 2..7);

$$R := 1 + X + C_2(X) + C_3(X) + C_4(X) + C_5(X) + C_6(X) + C_7(X)$$

> A := arbrer(R, 8);

$$\begin{aligned}
A := & X^7 + 2 E_2(X^3) + E_2(X^2) + X C_3(X^2) + X C_7(X) + X C_6(X) + X E_2(X) \\
& + X C_4(X) + X C_5(X) + X + 3 X E_2(X^3) + X E_2(X^2) + X C_3(X) \\
& + 5 E_2(X^4) + 3 X^6 + E_2(X) + X^5 + 28 X^8
\end{aligned}$$

> sge(A,8);

$$\frac{7}{4} x^5 + \frac{21}{5} x^6 + 11 x^7 + \frac{5}{6} x^4 + \frac{1}{2} x^2 + \frac{1}{2} x^3 + \frac{429}{14} x^8 + x$$

> sgef(A,8);

$$\frac{x}{1!} + \frac{x^2}{2!} + \frac{3x^3}{3!} + \frac{20x^4}{4!} + \frac{210x^5}{5!} + \frac{3024x^6}{6!} + \frac{55440x^7}{7!} + \frac{1235520x^8}{8!}$$

> sgt(A,8);

$$x + x^2 + x^3 + 2x^4 + 3x^5 + 6x^6 + 14x^7 + 34x^8$$

4.4.3 G -additive rooted trees

The command `arborgadd(G,d)` returns the molecular expansion up to degree d of the species a_G of G -additive rooted trees.

Example.

> G := E[2](X) + E[3](X);

$$G := E_2(X) + E_3(X)$$

> A := arborgadd(G,6);

$$\begin{aligned} A := & X + E_2(X) + E_3(X) + E_2(X)X + E_2(X)^2 + E_2(E_2(X)) + X E_3(X) \\ & + E_2(X)X^2 + 2 E_2(X) E_3(X) + 3 X E_2(X)^2 + 2 X E_2(E_2(X)) \\ & + X^2 E_3(X) + E_2(X)X^3 + 6 E_2(X)X E_3(X) \\ & + 2 E_2(X) E_2(E_2(X)) + 6 E_2(X)^2 X^2 + E_3(E_2(X)) + 2 E_2(X)^3 \\ & + E_2(E_3(X)) + E_2(E_2(X)X) + 2 X^2 E_2(E_2(X)) + X^3 E_3(X) \\ & + E_2(X)X^4 \end{aligned}$$

> sgt(A,6);

$$x + x^2 + 2x^3 + 4x^4 + 9x^5 + 23x^6$$

4.5 A procedure for computing the n^{th} root of a species

There is a procedure, `rac` (English: `nroot`), in *Devmol*, for computing the n^{th} root of a species. An n^{th} root of a species F is a species Φ such that $\Phi^n = F$. As in the case of formal power series, Φ is uniquely determined once its constant term Φ_0 has been given, satisfying $\Phi_0^n = F_0$, under the condition that F_0 is invertible in the ring $\mathbb{C}[V]$; and the algorithm for computing Φ

is the same as for formal power series. It makes use of Newton's binomial theorem. Let $F = F_0 + F_+$. Then we have

$$\begin{aligned}\Phi &= F^{1/n} \\ &= F_0^{1/n} \left(1 + \frac{F_+}{F_0}\right)^{1/n} \\ &= \Phi_0 \cdot \sum_{k \geq 0} \binom{1/n}{k} \left(\frac{F_+}{F_0}\right)^k.\end{aligned}\tag{26}$$

The command `rac(n,F,Phi_0,d)` returns the molecular expansion of (26) truncated at degree d . Observe that we could compute in the same manner any rational power F^α of F , for $\alpha \in \mathbb{Q}$.

Examples. 1. The following command sequence returns the square root of the species $A(2X)/(2X)$, of constant term 1, up to degree 6, where A denotes the species of rooted trees. It can be seen that the species $B = \sqrt{A(2X)/(2X)}$ satisfies the equation

$$B = 1 + E(XB^2).\tag{27}$$

In fact, B is the species of triangle-labelled oriented-edge-rooted 2-trees (see [5]).

```
> R := 1 + sum(E[k](X), k = 1..6);
```

$$R := 1 + E_1(X) + E_2(X) + E_3(X) + E_4(X) + E_5(X) + E_6(X)$$

```
> A := arborr(R,7):
```

```
> A2 := dev(subs(X = 2*X,A)/(2*X),6):
```

```
> rac(2,A2,1,6);
```

$$\begin{aligned}1 + 2 X E_2(X) + 2 X E_3(X) + 10 X^2 E_2(X) + 2 X E_4(X) + 43 X^3 E_2(X) \\ + 6 X E_2(X^2) + 10 X^2 E_3(X) + 18 X^2 E_2(X)^2 + 43 X^3 E_3(X) \\ + 22 X^2 E_2(X^2) + 201 X^4 E_2(X) + 10 X^2 E_4(X) + X E_5(X) \\ + 2 E_2(X^2) + 2 E_3(X^2) + 5 E_2(X^3) + 2 E_2(X E_2(X)) + X \\ + 6 E_3(X) X E_2(X) + 2 E_2(X) E_2(X^2) + 388 X^6 + E_4(X) \\ + E_5(X) + E_6(X) + E_2(X) + E_3(X) + 2 X^2 + 7 X^3 + 24 X^4 \\ + 3 X E_2(X)^2 + 95 X^5\end{aligned}$$

2. The species $P = P(X)$ of parenthesizations, introduced in Section 3.3,

is defined by the functional equation (25) or equivalently, by

$$P^2 - P + X = 0, \quad (28)$$

Solving this quadratic equation yields

$$P(X) = \frac{1 - \sqrt{1 - 4X}}{2}. \quad (29)$$

This expression is expanded up to degree 10 in the following lines. Anyone will recognize the familiar Catalan numbers.

```
> S := rac(2,1-4*X,1,10);
```

$$S := 1 - 2X - 2X^2 - 4X^3 - 10X^4 - 28X^5 - 84X^6 - 264X^7 \\ - 858X^8 - 2860X^9 - 9724X^{10}$$

```
> P(X) := (1 - S)/2;
```

$$P(X) := X + X^2 + 2X^3 + 5X^4 + 14X^5 + 42X^6 + 132X^7 + 429X^8 \\ + 1430X^9 + 4862X^{10}$$

4.6 A procedure for computing the inverse under substitution

The inverse under substitution of a species $\Phi = \Phi(X)$ is defined whenever $\Phi(0) = 0$ and $\Phi'(0) = 1$, i. e. when Φ is of the form $\Phi = X + \Phi_{\geq 2}$. The inverse of such a species is then the species Ψ such that $\Phi \circ \Psi = X$. The procedure `invsubst` (English: `invsubs`) computes the inverse under substitution of a species, up to the given degree.

Examples. 1. We compute the inverse under substitution up to degree 5 of the species $X + E_2(X) + \dots + E_5(X)$. Note that this inverse is equal to the inverse of $E_+ = E - 1$ up to the same degree.

```
> Phi := sum(E[k](X), k = 1..5);
```

$$\Phi := E_1(X) + E_2(X) + E_3(X) + E_4(X) + E_5(X)$$

> invsubst(Phi, 5);

$$\begin{aligned}
& X - E_2(X) - E_3(X) + X E_2(X) - E_4(X) + E_2(E_2(X)) + X E_3(X) \\
& - X^2 E_2(X) - E_5(X) + X E_4(X) + E_2(X) E_3(X) - X^2 E_3(X) \\
& - E_2(X)^2 X + X^3 E_2(X)
\end{aligned}$$

2. The equation (28) shows that the species $P(X)$ is the substitutional inverse of $X - X^2$. This is verified here up to degree 10.

> Phi := X - X^2;

$$\Phi := X - X^2$$

> invsubst(Phi, 10);

$$\begin{aligned}
& X + X^2 + 2 X^3 + 5 X^4 + 14 X^5 + 42 X^6 + 132 X^7 + 429 X^8 + 1430 X^9 \\
& + 4862 X^{10}
\end{aligned}$$

3. In this example, we verify, up to degree 6, that the species A of rooted trees satisfies the combinatorial equation $A = (X E(-X))^{(-1)}$.

> EE := 1 + sum(E[k](X), k = 1..5);

$$EE := 1 + E_1(X) + E_2(X) + E_3(X) + E_4(X) + E_5(X)$$

> AA := arborrh(EE, 6);

$$\begin{aligned}
& X + X^2 + X E_2(X) + X^3 + X E_3(X) + 2 X^4 + X^2 E_2(X) + X E_4(X) \\
& + 3 X^3 E_2(X) + X E_2(X^2) + 3 X^5 + X^2 E_3(X) + 6 X^6 + X E_5(X) \\
& + 2 X^2 E_2(X^2) + 3 X^3 E_3(X) + X^2 E_4(X) + 6 X^4 E_2(X) \\
& + X^2 E_2(X)^2
\end{aligned}$$

> FF := dev(X*subs(X=-X, EE), 6);

$$\begin{aligned}
FF := & X - X^2 - X E_2(X) + X^3 - X E_3(X) + 2 X^2 E_2(X) - X^4 - X E_4(X) \\
& + 2 X^2 E_3(X) + X E_2(X)^2 - 3 X^3 E_2(X) + X^5 - X E_5(X) \\
& + 2 X^2 E_4(X) + 2 X E_3(X) E_2(X) - 3 X^3 E_3(X) - 3 X^2 E_2(X)^2 \\
& + 4 X^4 E_2(X) - X^6
\end{aligned}$$

> BB := invsubst(FF, 6);

$$\begin{aligned}
BB := & X + X^2 + X E_2(X) + X^3 + X E_3(X) + 2 X^4 + X^2 E_2(X) + X E_4(X) \\
& + 3 X^3 E_2(X) + X E_2(X^2) + 3 X^5 + X^2 E_3(X) + 6 X^6 + X E_5(X) \\
& + 2 X^2 E_2(X^2) + 3 X^3 E_3(X) + X^2 E_4(X) + 6 X^4 E_2(X) \\
& + X^2 E_2(X)^2
\end{aligned}$$

> AA - BB;

0

4.7 Degree and table procedures

This category consists essentially of ‘interface’ procedures related to the degree of a molecular species and to the homogeneous components of a species. These are the procedures `degre`, `mdegre`, `coeffm` (English: `mcoeff`), `phom` (`homcomp`), `tablephom` (`homcomptable`), `affichertable` (`printtable`), `sphom` (`homcompsum`), as well as `tablearberr` (`rrtreestable`), `tablearbrer` (`rtreestable`) and `tablearborgadd` (`gaddrtreestable`).

The command `degre(M)` returns the total degree of the molecular species M . In case of a multisort molecular species, the command `mdegre(M)` can also be used, the result being a Maple table. The command `coeffm(F,M)` yields the coefficient (multiplicity) of the molecular species M in the species F . Here are some examples.

> `degre(E[3](X*C[4](X)));`

15

> `mdegre(E[3](X*C[4](Y^2*Z)));`

resultat

> `print(%);`

`table([X = 3, Y = 24, Z = 12])`

> `Phi:=1+X^2+23*E[3](X*C[4](X))+6*E[3](C[5](X))+E[7](X);`

$$\Phi := 1 + X^2 + 23E_3(XC_4(X)) + 6E_3(C_5(X)) + E_7(X)$$


```
> coeffm(Phi,E[3](X*C[4](X)));
23
```

The command `phom(F,d)` returns the homogeneous part of degree d of the species F , while `tablephom(F)` stores in a Maple table the degrees which occur in F and the corresponding homogeneous components. This is particularly useful when the species F is big. Conversely, the procedure `sphom` converts a table into the sum of its entries and hence returns the sum of the homogeneous parts in a table of the form `tablephom(F)`. The procedure `affichertable` is a practical alternative to the Maple command `print` for displaying a table.

```
> phom(Phi,15);
```

$$23 E_3(X C_4(X)) + 6 E_3(C_5(X))$$

```
> T := tablephom(Phi);
```

$T := tph$

```
> print(T);
```

```
table([0 = 1, 2 = X^2, 7 = E_7(X), 15 = 23 E_3(X C_4(X)) + 6 E_3(C_5(X))])
```

```
> affichertable(T);
```

$$\begin{array}{c} 0 \\ 1 \\ 2 \\ X^2 \\ 7 \\ E_7(X) \\ 15 \\ 23 E_3(X C_4(X)) + 6 E_3(C_5(X)) \end{array}$$

```
> sphom(T);
```

$$1 + X^2 + 23 E_3(X C_4(X)) + 6 E_3(C_5(X)) + E_7(X)$$

Finally, the procedures `tablearberr`, `tablearbrer` and `tablearborgadd` are simply the composition of `tablephom` with the procedures `arberr`, `arbrer`

and `arborgadd` for enriched trees and rooted trees.

4.8 Miscellaneous

There are two other useful *Devmol* procedures, `derpart` (English: `partder`) and `grouper` (`mcollect`), which are illustrated in the following section. The procedure `derpart` computes the partial derivative of a species with respect to a given sort variable. The two basic facts here are the formulas

$$\frac{\partial}{\partial X} E_n(X) = E_{n-1}(X) \quad \text{and} \quad \frac{\partial}{\partial X} C_n(X) = X^{n-1}. \quad (30)$$

Finally, the procedure `grouper` collects similar terms in a molecular expansion, that is terms having the same molecule but various weights.

5 Two applications

In this last section, we give two examples of applications of *Devmol*. The first one is the computation of the molecular expansion of the species A_w of rooted trees weighted by a leaf counter t . The second application is a procedure for the molecular expansion of the species $C_{\mathcal{B}}$ of connected (simple) graphs, all of whose blocks are in a given class \mathcal{B} of 2-connected graphs, for a generic \mathcal{B} . This procedure is then illustrated in the case of the species Hu_w of weighted Husimi graphs presented as an example in Section 1.2.

5.1 Rooted trees weighted with a leaf counter.

Let us define the weight $w(\alpha)$ of a rooted tree α as $t^{l(\alpha)}$, where $l(\alpha)$ denotes the number of leaves of α . By convention, the root of a rooted tree is a leaf only in the case of the single vertex rooted tree. The species of rooted trees weighted in this manner will be denoted by A_w . A slight modification of equation (5) leads to the functional equation

$$\begin{aligned} A_w &= X(t-1) + X E(A_w) \\ &= tX + X E_+(A_w). \end{aligned} \quad (31)$$

It is thus possible to compute iteratively the molecular expansion of the species A_w as explained in Section 3.1, up to a given degree d . The procedure `lwrtrees(d)` which performs this computation is presented below.

```

lwrtrees := proc(n)
local k, Awl, Ek;
with(devmol);
Awl := t * X;
Ek := 0;
for k to n - 1 do
Ek := Ek + Ek(X);
Awl := t * X + expand(X * devmol : -dev(subs(X = Awl, Ek), k))
end do;
devmol : -grouper(Awl)
end proc

```

Here is the result of this computation up to degree 8.

```
> affichertable(tablephom(lwrtrees(8)));
```

$$\begin{array}{c}
1 \\
t X \\
2 \\
t X^2 \\
3 \\
t^2 X E_2(X) + t X^3 \\
4 \\
t^2 X^2 E_2(X) + t^3 X E_3(X) + (t + t^2) X^4 \\
5 \\
(t + 2t^2) X^5 + (2t^3 + t^2) X^3 E_2(X) + t^3 X^2 E_3(X) + t^2 X E_2(X^2) \\
+ t^4 X E_4(X) \\
6 \\
t^4 E_2(X)^2 X^2 + t^4 X^2 E_4(X) + t^5 X E_5(X) + (t^3 + t^2) X^2 E_2(X^2) \\
+ (2t^4 + t^3) X^3 E_3(X) + (t^2 + 5t^3) X^4 E_2(X) \\
+ (t + 4t^2 + t^3) X^6
\end{array}$$

$$\begin{aligned}
& \begin{aligned}
& 7 \\
& 2t^5 E_2(X) X^2 E_3(X) + t^4 X E_2(X) E_2(X^2) + (t^2 + 9t^3 + 4t^4) X^5 E_2(X) \\
& \quad + 2t^4 E_2(X)^2 X^3 + t^5 X^2 E_5(X) + (2t^3 + t^2) X^3 E_2(X^2) \\
& \quad + (5t^4 + t^3) X^4 E_3(X) + (t^4 + 2t^5) X^3 E_4(X) + X t^2 E_2(X^3) \\
& \quad \quad + t^4 X E_2(X E_2(X)) + t^3 X E_3(X^2) + t^6 X E_6(X) \\
& \quad \quad \quad + (t + 6t^2 + 5t^3) X^7
\end{aligned} \\
& \begin{aligned}
& 8 \\
& t^5 X E_3(X) E_2(X^2) + 2t^6 E_2(X) X^2 E_4(X) + 3t^4 X^2 E_2(X) E_2(X^2) \\
& \quad + 5t^5 E_2(X) X^3 E_3(X) + (2t^4 + t + 9t^2 + 13t^3) X^8 \\
& \quad + (t^4 + t^5) X^2 E_2(X E_2(X)) + (t^3 + t^2) X^2 E_2(X^3) + t^7 X E_7(X) \\
& \quad \quad + t^6 X^2 E_6(X) + t^6 E_3(X)^2 X^2 + (t^5 + 2t^6) X^3 E_5(X) \\
& \quad \quad + (t^2 + 5t^3 + t^4) X^4 E_2(X^2) + (14t^3 + t^2 + 17t^4) X^6 E_2(X) \\
& \quad \quad + (t^3 + 9t^4 + 4t^5) X^5 E_3(X) + (4t^4 + 4t^5) E_2(X)^2 X^4 \\
& \quad \quad \quad + (5t^5 + t^4) X^4 E_4(X) + (t^3 + t^4) X^2 E_3(X^2)
\end{aligned}
\end{aligned}$$

5.2 Connected graphs with blocks in a given class \mathcal{B}

Let \mathcal{B} be a class of 2-connected graphs. A $C_{\mathcal{B}}$ -graph is defined as a connected graph all of whose blocks (2-connected components) are in \mathcal{B} . The species of $C_{\mathcal{B}}$ -graphs is denoted by $C_{\mathcal{B}}$. Here are some examples for various choices of \mathcal{B} :

1. If $\mathcal{B} = \mathcal{B}_a$, the class of *all* 2-connected graphs, then $C_{\mathcal{B}} = \mathcal{C}$, the species of (all) connected graphs.
2. If $\mathcal{B} = K_2$, the class of "edges", then $C_{\mathcal{B}} = a$, the species of (unrooted, free) trees.
3. If $\mathcal{B} = \{P_m, m \geq 2\}$, where P_m denotes the class of size- m polygons (by convention, $P_2 = K_2$), then $C_{\mathcal{B}} = \text{Ca}$, the species of cacti. A *cactus* can also be defined as a connected graph in which no edge lies in more than one cycle. Figure 7, a), represents a typical cactus.
4. If $\mathcal{B} = K_3 = P_3$, the class of "triangles", then $C_{\mathcal{B}} = \delta$, the class of triangular cacti.
5. If $\mathcal{B} = \{K_n, n \geq 2\}$, the family of complete graphs, then $C_{\mathcal{B}} = \text{Hu}$, the species of *Husimi graphs*. A Husimi graph is shown in Figure 2.

6. If $\mathcal{B} = \{C_n, n \geq 2\}$, the family of oriented cycles, then $C_{\mathcal{B}} = \text{Oc}$, the species of *oriented cacti*. These structures were introduced by C. Springer [20] in 1996. Although directed graphs are involved here, the functional equations (32) and (33) given below are still valid. Figure 7, b) shows a typical oriented cactus.




Figure 7: a) a typical cactus, b) a typical oriented cactus

Let $C_{\mathcal{B}}^{\bullet}$ denote the species of rooted (at a vertex) $C_{\mathcal{B}}$ -graphs. Then the following two equations (see [2], [14]) can be used to compute the molecular expansion of the species $C_{\mathcal{B}}$.

$$C_{\mathcal{B}}^{\bullet} = X E(\mathcal{B}'(C_{\mathcal{B}}^{\bullet})), \quad (32)$$

$$C_{\mathcal{B}} = C_{\mathcal{B}}^{\bullet} + \mathcal{B}(C_{\mathcal{B}}^{\bullet}) - C_{\mathcal{B}}^{\bullet} \mathcal{B}'(C_{\mathcal{B}}^{\bullet}). \quad (33)$$

Here, \mathcal{B}' denotes the derivative of the species \mathcal{B} . Observe that equation (32) is of the form $Y = X R(Y)$, with $R = E(\mathcal{B}')$, so that the procedure for R -enriched rooted trees can be used. Moreover, equations (32) and (33) are also valid for weighted $C_{\mathcal{B}}$ -graphs, whenever the weight function w is multiplicative on the 2-connected components. See [14] for more details.

The procedure `CBgraphs` takes the class \mathcal{B} of blocks and the degree d as arguments. It computes the molecular expansion of the species of $C_{\mathcal{B}}$ -graphs up to degree d .

```

CBgraphes := proc(B, n)
local EE, Bprime, R, GP, resultat;
with(devmol);
EE := 1 + sum(Ek(X), k = 1..n - 1);
Bprime := devmol : -derpart(B, X);
R := devmol : -dev(subs(X = Bprime, EE), n - 1);
GP := devmol : -arborr(R, n);
resultat := devmol : -dev(subs(X = GP, X + B - X * Bprime), n)
end proc

```

In what follows, *devmol* computes the molecular expansion, up to degree 7, of the weighted species Oc_w of oriented cacti, where the weight $w(c)$ of an oriented cactus c is defined by $w(c) = y_2^{n_2} y_3^{n_3} \cdots$, where n_k is the number of cycles of c of size k . This is the case of C_B -graphs with $B = B_w$, the species of weighted oriented cycles of length at least 2, where the weight of a cycle of length k is y_k .

```
> n := 7;
```

```
n := 7
```

```
> ajoutvv(seq(y[k], k = 1..n));
```

```
{t, y1, y2, y3, y4, y5, y6, y7}
```

```
> B := sum(y[k]*C[k](X), k = 2..n);
```

$$B := y_2 C_2(X) + y_3 C_3(X) + y_4 C_4(X) + y_5 C_5(X) + y_6 C_6(X) + y_7 C_7(X)$$

```
> CBW := CBgraphes(B,n):
```

> affichertable(tablephom(CBW));

$$\begin{aligned}
& 1 \\
& X \\
& 2 \\
& y_2 E_2(X) \\
& 3 \\
& y_2^2 X E_2(X) + y_3 C_3(X) \\
& 4 \\
& y_2 y_3 X^4 + y_2^3 X E_3(X) + y_2^3 E_2(X^2) + y_4 C_4(X) \\
& 5 \\
& (y_3 y_2^2 + y_2^4) X^3 E_2(X) + (y_2^4 + y_3^2) X E_2(X^2) + y_2^4 X E_4(X) \\
& \quad + (y_4 y_2 + 2 y_3 y_2^2) X^5 + y_5 C_5(X) \\
& 6 \\
& y_2^5 E_2(X E_2(X)) + (y_2^5 + y_3^2 y_2 + y_4 y_2^2) E_2(X^3) \\
& + (y_3 y_2^3 + y_2^5) X^3 E_3(X) + (y_4 y_2^2 + 3 y_3 y_2^3 + y_2^5) X^4 E_2(X) \\
& \quad + (y_2^5 + y_3^2 y_2) X^2 E_2(X^2) + y_3 y_2^3 C_3(X^2) + y_2^5 X E_5(X) \\
& \quad + (2 y_3^2 y_2 + 2 y_4 y_2^2 + 4 y_3 y_2^3 + y_5 y_2 + y_3 y_4) X^6 + y_6 C_6(X) \\
& 7 \\
& (7 y_4 y_3 y_2 + y_3^3 + y_2^6 + y_5 y_3 + y_6 y_2 + 3 y_5 y_2^2 + 6 y_4 y_2^3 + 9 y_3 y_2^4 + 9 y_3^2 y_2^2) X^7 \\
& \quad + (2 y_3^2 y_2^2 + y_5 y_2^2 + 8 y_3 y_2^4 + 4 y_4 y_2^3 + 2 y_2^6) X^5 E_2(X) + y_3 y_2^4 X^3 E_2(X)^2 \\
& \quad + (y_3 y_2^4 + y_3^2 y_2^2) X^3 E_2(X^2) + (y_4 y_2^3 + 3 y_3 y_2^4 + y_2^6) X^4 E_3(X) + \\
& \quad \quad (y_3 y_2^4 + y_2^6) X^3 E_4(X) + (3 y_3^2 y_2^2 + y_4^2 + y_2^6) X E_2(X^3) \\
& \quad \quad + (y_2^6 + y_3^2 y_2^2) X E_2(X) E_2(X^2) + y_2^6 X E_2(X E_2(X)) \\
& \quad + (y_3^3 + y_2^6) X E_3(X^2) + y_2^6 E_3(X) E_2(X) X^2 + y_2^6 X E_6(X) + y_7 C_7(X)
\end{aligned}$$

6 Availability of Devmol

The current version is Devmol 2.2. It is constructed as a Maple module. The basic version is in French but an English version, where the English aliases of the exported procedures are implemented, is also available.

It is freely available at the Web site of LaCIM: www.lacim.uqam.ca.

6.1 Glossary

Here is a short dictionary, from French to English, of the exported procedures of Devmol.

affichertable = printtable
ajoutvs = addsv
ajoutvv = addwv
arborgadd = gaddrtrees
arborr = rrtrees
arbrer = rtrees
coeffm = mcoeff
derpart = partder
grouper = mcollect
invsbst = invsubs
phom = homcomp
rac = nroot
sge = egs
sgef = fegs
sgt = tgs
sphom = homcompsum
tablearborgadd = gaddrtreestable
tablearborr = rrtreestable
tablearbrer = rtreestable
tablephom = homcomptable

References

- [1] Pierre Auger, Gilbert Labelle and Pierre Leroux, *Combinatorial addition formulas and applications*, Advances in Applied Mathematics, 28 (2002), 302–342.
- [2] François Bergeron, Gilbert Labelle and Pierre Leroux, *Combinatorial Species and Tree-Like Structures*, Cambridge University Press, 1998, 457 p. Vol. 67 of the series Encyclopedia of Mathematics and its Applications.
- [3] Miklós Bóna, Michel Bousquet, Gilbert Labelle and Pierre Leroux, *Enumeration of m -ary cacti*, Advances in Applied Mathematics, 24, 2000, 22–56.
- [4] Yves Chiricota, *Structures combinatoires et calcul symbolique*, PhD Thesis, UQAM, 1992, Publications du LaCIM, Vol. 12, 1993.
- [5] Tom Fowler, Ira Gessel, Gilbert Labelle and Pierre Leroux, *The specification of 2-tress*, Advances in Applied Mathematics, 28, 2002, 145–168.
- [6] Frank Harary, *Graph Theory*, Addison-Wesley, 1969, 274 p.
- [7] Frank Harary and Edgar Palmer. *Graphical Enumeration*, Academic Press, New York and London, 1973, 271 p.
- [8] Kodi Husimi. *Note on Mayers' theory of cluster integrals*, The Journal of Chemical Physics, 18, 1950, 682–684.
- [9] André Joyal, *Une théorie combinatoire des séries formelles*, Advances in Mathematics, 42, 1981, 1–82.
- [10] Gilbert Labelle, *Counting asymmetric enriched trees*, Journal of Symbolic Computation, 14, 1992, 211–242.
- [11] Gilbert Labelle, Cédric Lamathe, Pierre Leroux, *Énumération des 2-arbres k -gonaux*, Proceedings of the conference Mathematics and Computer Science: Algorithms, Trees, Combinatorics and Probabilities, Versailles, September 16–19, 2002, in *Mathematics and Computer Science II*, B. Chauvin, P. Flajolet, D. Gardy, and A. Mokkaedem, ed., Birkhäuser, Basel, 2002, 95–109.

- [12] Gilbert Labelle, Cédric Lamathe and Pierre Leroux, *A classification of plane and planar 2-trees*, Theoretical Computer Science, To appear.
- [13] Pierre Leroux, *Methoden der Anzahlbestimmung für einige Klassen von Graphen*, Bayreuther Mathematische Schriften, 26, 1988, 1–36.
- [14] Pierre Leroux, *Enumerative problems inspired by Mayer's theory of cluster integrals*, Plenary talk, FPSAC03 Conference, Linköping University (Sweden), June 23 - 27, 2003.
- [15] Pierre Leroux and Brahim Miloudi, *Généralisations de la formule d'Otter*, Annales des Sciences Mathématiques du Québec, 16, 1992, 53–80.
- [16] Mélanie Nadeau, *Graphes de Husimi, cactus, et mécanique statistique*. Mémoire de maîtrise, UQAM, 2002.
- [17] Robert Z. Norman, *On the number of linear graphs with given blocks*, Dissertation, University of Michigan, 1954.
- [18] Richard Otter, *The number of trees*, Annals of Mathematics, 49, 1948, 583–599.
- [19] Robert Robinson, *Enumeration of non-separable graphs*, Journal of Combinatorial Theory, 9, 1970, 327–356.
- [20] Colin Springer, *Factorizations, trees, and cacti*, Proceedings of the Eighth International Conference on Formal Power Series and Algebraic Combinatorics (FPSAC), University of Minnesota, 1996, 427–438.